
Motivation

What if a program transformation could:

• ensure that every function call is a tail call and the stack is explicit,
so the code is no longer really recursive, but iterative;

• make the evaluation order explicit in the code,
so that it does not depend on the ambient strategy (CBN / CBV);

• eliminate the apparent redundancy between calls and returns,
by exploiting solely function calls – functions never return!

• suggest extending the λ-calculus with control operators?

The continuation-passing style transformation does all this.

Motivation

Steele, RABBIT: a compiler for SCHEME, 1978.

http://repository.readscheme.org/ftp/papers/ai-lab-pubs/AITR-474.pdf

From a direct-style interpreter down to an abstract machine

From recursive traversal down to iterative traversal with link inversion

A direct-style interpreter

Recall our environment-based interpreter for call-by-value λ-calculus:

let rec eval (e : cenv) (t : term) : cvalue =
match t with
| Var x ->

lookup e x
| Lam t ->

Clo (t, e)
| App (t1, t2) ->

let cv1 = eval e t1 in
let cv2 = eval e t2 in
let Clo (u1, e’) = cv1 in
eval (cv2 :: e’) u1

This is an OCaml transcription, without a fuel parameter.

A continuation-passing style interpreter

Instead of returning a value,

let rec eval (e : cenv) (t : term) : cvalue =
...

let’s pass this value to a continuation that we get as an argument:

let rec evalk (e : cenv) (t : term) (k : cvalue -> ’a) : ’a =
...

Exercise (in class): write evalk. (See EvalCBVExercise.)

https://gitlab.inria.fr/fpottier/mpri-2.4-public/tree/master/ocaml/EvalCBVExercise.ml

A continuation-passing style interpreter

let rec evalk (e : cenv) (t : term) (k : cvalue -> ’a) : ’a =
match t with
| Var x ->

k (lookup e x)
| Lam t ->

k (Clo (t, e))
| App (t1, t2) ->

evalk e t1 (fun cv1 ->
evalk e t2 (fun cv2 ->
let Clo (u1, e’) = cv1 in
evalk (cv2 :: e’) u1 k))

Instead of returning a value, pass it to k.

Instead of sequencing computations via let, nest continuations.

A continuation-passing style interpreter

To run the interpreter, start it with the identity continuation:

let eval (e : cenv) (t : term) : cvalue =
evalk e t (fun cv -> cv)

Correctness of the CPS interpreter

The continuation-passing style interpreter is “obviously” correct.

Exercise: define evalk in Coq (with fuel) and prove it equivalent
to the direct-style interpreter: evalk n e t k = k (eval n e t).

Properties of the interpreter

What is special about this interpreter?

• Every call of evalk to itself is a tail call.
• Every call of evalk to a continuation is a tail call.

Tail calls

A call g x is a tail call if it is the “last thing” that the calling function does...

More formally,

v ::= x | λx .tt values
tt ::= terms in tail position
| v
| nt nt – a tail call
| let nt in tt
| if nt then tt else tt

nt ::= terms not in tail position
| v
| nt nt – an ordinary call
| let nt in nt
| if nt then nt else nt

Verified tail calls

OCaml allows us to verify that these are indeed tail calls:

let rec evalk (e : cenv) (t : term) (k : cvalue -> ’a) : ’a =
match t with
| Var x ->

(k[@tailcall]) (lookup e x)
| Lam t ->

(k[@tailcall]) (Clo (t, e))
| App (t1, t2) ->

(evalk[@tailcall]) e t1 (fun cv1 ->
(evalk[@tailcall]) e t2 (fun cv2 ->
let Clo (u1, e’) = cv1 in
(evalk[@tailcall]) (cv2 :: e’) u1 k))

A nice feature (though with somewhat ugly syntax).

Properties of the interpreter

Tail calls are compiled by OCaml to jumps.

Thus, tail-recursive functions are compiled by OCaml to loops.

Steele, Lambda: the ultimate GOTO, 1977.

Thus, the CPS interpreter is not truly recursive: it is iterative.

It uses constant space on OCaml’s implicit stack.

Wait! Does the interpreter really not need a stack any more?

• Of course it does need a stack.
• The continuation, allocated in the OCaml heap, serves as a stack.

https://dspace.mit.edu/bitstream/handle/1721.1/5753/AIM-443.pdf

A defunctionalized CPS interpreter

To better see the structure of the continuation,
let us defunctionalize the CPS interpreter.

Reynolds, Definitional interpreters
for programming languages, 1972 (1998).

Reynolds, Definitional interpreters revisited, 1998.

https://doi.org/10.1023/A:1010027404223
https://doi.org/10.1023/A:1010027404223
https://doi.org/10.1023/A:1010075320153

Defunctionalization (reminder)

Steps:

• Identify the sites where closures are allocated,
that is, where anonymous functions are built.

• Compute, at each site, the free variables of the anonymous function.
• Introduce an algebraic data type of closures.
• Transform the code:

• replace anonymous functions with constructor applications,
• replace function applications with calls to apply,
• and define apply.

Exercise (in class): defunctionalize the CPS interpreter. (EvalCBVExercise.)

https://gitlab.inria.fr/fpottier/mpri-2.4-public/tree/master/ocaml/EvalCBVExercise.ml

A defunctionalized CPS interpreter

There are three sites where an anonymous continuation is built.

We name them and compute their free variables.

This leads to the following algebraic data type of continuations:

type kont =
| AppL of { e: cenv; t2: term; k: kont }
| AppR of { cv1: cvalue; k: kont }
| Init

What data structure is this?

A linked list. A heap-allocated stack.

In fact, it is a (call-by-value) evaluation context:

E ::= E t2[e] | v1 E | [ ]

A defunctionalized CPS interpreter

We transform the interpreter’s main function:

let rec evalkd (e : cenv) (t : term) (k : kont) : cvalue =
match t with
| Var x ->

apply k (lookup e x)
| Lam t ->

apply k (Clo (t, e))
| App (t1, t2) ->

evalkd e t1 (AppL { e; t2; k })

To evaluate t1 t2, the interpreter pushes information on the stack,
then jumps straight to evaluating t1.

A defunctionalized CPS interpreter

apply interprets continuations as functions of values to values:

and apply (k : kont) (cv : cvalue) : cvalue =
match k with
| AppL { e; t2; k } ->

let cv1 = cv in
evalkd e t2 (AppR { cv1; k })

| AppR { cv1; k } ->
let cv2 = cv in
let Clo (u1, e’) = cv1 in
evalkd (cv2 :: e’) u1 k

| Init ->
cv

It pops the top stack frame and decides what to do, based on it.

A defunctionalized CPS interpreter

To run the interpreter, start it with the identity continuation:

let eval e t =
evalkd e t Init

An abstract machine

We have reached an abstract machine, a simple iterative interpreter
which maintains a few data structures:

• a code pointer: the term t,
• an environment e,
• a stack, or continuation k.

In fact, we have mechanically rediscovered the CEK machine.

Felleisen and Friedman,
Control operators, the SECD machine, and the λ-calculus, 1987.

Sig Ager, Biernacki, Danvy and Midtgaard,
A Functional Correspondence between Evaluators

and Abstract Machines, 2003.

http://www.cs.tufts.edu/~nr/cs257/archive/matthias-felleisen/cesk.pdf
http://www.brics.dk/RS/03/13/BRICS-RS-03-13.pdf
http://www.brics.dk/RS/03/13/BRICS-RS-03-13.pdf

Re-discovering other abstract machines

Exercise: start with a call-by-name interpreter and follow an analogous
process to rediscover Krivine’s machine.

The solution is in EvalCBNCPS.

There once was a man named Krivine
Who invented a wond’rous machine.
It pushed and it popped
On abstractions it stopped;
That lean mean machine from Krivine.

— Mitchell Wand

Krivine, A call-by-name lambda-calculus machine, (1985) 2007.

https://gitlab.inria.fr/fpottier/mpri-2.4-public/tree/master/ocaml/EvalCBNCPS.ml
https://link.springer.com/journal/10990/20/3/page/1
https://www.irif.fr/~krivine/articles/lazymach.pdf

A type of binary trees

Consider a simple type of binary trees:

type tree =
| Leaf
| Node of { data: int; left: tree; right: tree }



MPRI 2.4
CPS

François
Pottier

Examples
Interpreter

Traversal

Formulations

Soundness

Remarks

25/94

Direct-style traversal

Suppose we wish to perform a postfix tree traversal:

let rec walk (t : tree) : unit =
match t with
| Leaf ->

()
| Node { data; left; right } ->

walk left;
walk right;
printf "%d\n" data

This is recursive code in direct style.

Neither of the recursive calls is a tail call.

CPS traversal

Now suppose we wish to make the code iterative. Swoop, CPS!

let rec walkk (t : tree) (k : unit -> ’a) : ’a =
match t with
| Leaf ->

k()
| Node { data; left; right } ->

walkk left (fun () ->
walkk right (fun () ->
printf "%d\n" data;
k()))

The traversal is initiated with an identity continuation:

let walk t =
walkk t (fun t -> t)

CPS traversal, defunctionalized

Next, we might wish to make the stack an explicit data structure.

Swoop, defunctionalization!

The type of defunctionalized continuations:

type kont =
| Init
| GoneL of { data: int; tail: kont; right: tree }
| GoneR of { data: int; tail: kont }

CPS traversal, defunctionalized

The main function is a loop that walks down the leftmost branch
while pushing information onto the stack:

let rec walkkd (t : tree) (k : kont) : unit =
match t with
| Leaf ->

apply k ()
| Node { data; left; right } ->

walkkd left (GoneL { data; tail = k; right })

Think of the stack as Ariadne’s thread.

CPS traversal, defunctionalized

The apply function comes back up out of a child.

and apply k () =
match k with
| Init ->

()
| GoneL { data; tail; right } ->

walkkd right (GoneR { data; tail })
| GoneR { data; tail } ->

printf "%d\n" data;
apply tail ()

It pops information off the stack so as to decide what to do.

When coming out of a left child, go down into its right sibling.

When coming out of a right child, go further up.

And now, for something a little

UNEXPECTED and WILD.

A CRAZY HACK.

Recycling

When we allocate a GoneR continuation,
we drop a GoneL continuation at the same time.

Inded, here, continuations are linear. They are used exactly once.

| GoneL { data; tail; right } ->
walkkd right (GoneR { data; tail })

This suggests that the memory block could be recycled (re-used).

More recycling

When we allocate a GoneL continuation,
a Node goes temporarily unused at the same time.

This node won’t be accessed until this GoneL frame
first is changed to GoneR then is popped off the stack.

| Node { data; left; right } ->
walkkd left (GoneL { data; tail = k; right })

This suggests that the memory block could be recycled, too,
provided we restore it when we are done with it.

A tree is a continuation is a tree

In OCaml, the type of a memory block cannot be changed over time.

Thus, recycling tree nodes as stack frames, and vice-versa,
requires trees and continuations to have the same type.

Uh?

A tree is a continuation is a tree

Could we disguise a continuation as a tree?

In other words, could a stack frame fit in a tree node?

type kont =
| Init
| GoneL of { data: int; tail: kont; right: tree }
| GoneR of { data: int; tail: kont }

type tree =
| Leaf
| Node of { data: int; left: tree; right: tree }

Yes, kind of.

We just need one extra bit of storage per tree node,
so as to distinguish GoneL and GoneR.

A tree is a continuation is a tree

Add one “status” bit per tree node. Make nodes mutable.

type status = GoneL | GoneR
type mtree = Leaf | Node of {

data: int; mutable status: status;
mutable left: mtree; mutable right: mtree

}
type mkont = mtree

Tree records and continuation records occupy the same space in memory.

Thus, a tree record can be turned into a continuation record, and back!

By convention, in a “tree” record, the status field is GoneL.

In a “continuation” record,

• either status is GoneL and the left field stores tail;
• or status is GoneR and the right field stores tail.

CPS traversal with link inversion

Instead of allocating a GoneL continuation,
we now change the tree record to a continuation record:

let rec walkkdi (t : mtree) (k : mkont) : unit =
match t with
| Leaf ->

apply k t
| Node ({ left; _ } as n) ->

(* Change this tree to a [GoneL] continuation. *)
assert (n.status = GoneL);
n.left (* n.tail *) <- k;
walkkdi left (t : mkont)

The left field is overwritten, which is scary! We must restore it later.

We find that, in every call to walkkdi t k and apply k t,
k is the parent of t in the tree.

CPS traversal with link inversion

The rest of the code, in its horrific glory:

and apply (k : mkont) (child : mtree) : unit =
match k with
| Leaf -> ()
| Node ({ status = GoneL; left = tail; right; _ } as n) ->

n.status <- GoneR; (* update continuation! *)
n.left <- child; (* restore orig. left child! *)
n.right (* n.tail *) <- tail;
walkkdi right k

| Node ({ data; status = GoneR; right = tail; _ } as n) ->
printf "%d\n" data;
n.status <- GoneL; (* change back to a tree! *)
n.right <- child; (* restore orig. right child! *)
apply tail (k : mtree)

This code runs in constant space. Look Ma, no stack! (Uh?)

CPS traversal with link inversion

More accurately, the stack is stored in the tree itself, by reversing pointers.

This hack technique is known as link inversion.

It was invented for use in garbage collectors, which must traverse the heap
without requiring a huge stack.

We have re-discovered it via the idea of allocating continuations in place.

Schorr and Waite, An efficient machine-independent procedure for
garbage collection in various list structures, 1967.

Hubert and Marché, A case study of C source code verification: the
Schorr-Waite algorithm, 2005.

Sobel and Friedman, Recycling continuations, 1998.

https://www.cs.purdue.edu/homes/hosking/690M/p501-schorr.pdf
https://www.cs.purdue.edu/homes/hosking/690M/p501-schorr.pdf
https://www.lri.fr/~marche/hubert05sefm.ps
https://www.lri.fr/~marche/hubert05sefm.ps
https://www.cs.indiana.edu/hyplan/dfried/rc.ps

CPS traversal with link inversion

“Kids, do not try this at home”: this idea is complicated and expensive.

(The OCaml GC imposes a write barrier: write operations are slow.)

Exercise: Extend the code to deal with graphs, where there can be sharing
and cycles. (Use a mark bit in every node.)

Formulations of the CPS transformation

There are many variants of the CPS transformation,
and sometimes many formulations of a single variant.

Let us begin with the simplest formulation: Fischer and Plotkin’s.

Fischer, Lambda-Calculus Schemata, (1972) 1993.

Plotkin, Call-by-name, call-by-value and the λ-calculus, 1975.

https://dx.doi.org/10.1007/BF01019461
https://homepages.inf.ed.ac.uk/gdp/publications/cbn_cbv_lambda.pdf

Definition of the CBV CPS transformation

A term is translated to a function of a continuation k to an answer.

JxK =

λk . k x

Jλx .tK = λk . k (λx .JtK)

Jt1 t2K = λk . Jt1K (λx1. Jt2K (λx2. x1 x2 k ))

Jlet x = t1 in t2K = λk . Jt1K (λx . Jt2K k )

A function λx .t is translated to a function of two arguments λx .λk ..

Definition of the CBV CPS transformation

One avoids some redundancy by distinguishing the translation of terms JtK
and the translation of values LvM.

LxM = x

Lλx .tM = λx .JtK

JvK = λk . k LvM

Jt1 t2K = λk . Jt1K (λx1. Jt2K (λx2. x1 x2 k ))

Jlet x = t1 in t2K = λk . Jt1K (λx . Jt2K k )

Indifference

In a transformed term, the right-hand side of every application is a value.

Therefore, its execution is indifferent to the choice
of a call-by-name or call-by-value evaluation strategy.

In other words, evaluation order is fully explicit in a transformed term.

CPS can serve as an encoding of call-by-value into call-by-name.

Stacklessness

In a transformed term, every call is a tail call.

Therefore, reduction under a context is not required.

That is, execution does not require a stack.

We could (but won’t) give a (small-step, substitution-based) semantics
that takes indifference and stacklessness into account.

Exercise: Propose such a semantics. Prove that, when executing a
CPS-transformed term, it is equivalent to the standard semantics.

Effect of the transformation of types

How are types transformed?

A value of type T is translated to a value of type LTM.

A computation of type T is translated to a computation of type JTK.

LαM = α

LT1 → T2M =

LT1M→ JT2K

JTK = (LTM→ A)→ A

The type A , known as the answer type, is arbitrary and fixed.

One may take A to be the empty type 0. Then, JTK is ¬¬LTM. The CPS
transformation is known in logic as the double-negation translation.

Exercise (recommended): state and prove Type Preservation.

Effect of the transformation of types – refined

Could the transformation of types be made more precise in some sense?

JTK = (LTM→ A)→ A

Every transformed term is in fact answer-type polymorphic:

JTK = ∀A .(LTM→ A)→ A

Furthermore, every transformed term invokes its continuation once:

JTK = ∀A .(LTM→ A)( A

However, these properties are violated in the presence of control effects.

Thielecke, From control effects to typed continuation passing, 2003.

http://www.cs.bham.ac.uk/~hxt/research/effects.pdf

Administrative redexes

The translation presented so far is naïve.

It produces many “administrative” β-redexes.

E.g., in an application of a variable to a variable:

Jf xK = λk . JfK (λx1. JxK (λx2. x1 x2 k ))
= λk . (λk . k LfM) (λx1. (λk . k LxM) (λx2. x1 x2 k ))
= λk . (λk . k f) (λx1. (λk . k x) (λx2. x1 x2 k ))

=β λk . (λx1. (λk . k x) (λx2. x1 x2 k )) f
=β λk . (λk . k x) (λx2. f x2 k )
=β λk . (λx2. f x2 k ) x
=β λk . f x k

This is inefficient: one function call is translated to five function calls!

Semantic preservation

Plotkin (1975) proved semantic preservation,
based on a small-step simulation diagram.

This proof is complicated by the presence of administrative reductions.

A simpler approach is to use big-step semantics in the hypothesis:

Lemma (Semantic Preservation)

If t ↓cbv v and if w is a value, then JtK w −→?
cbv w LvM.

One should prove, in addition, that divergence is preserved.

Exercise (recommended): prove this lemma.

https://homepages.inf.ed.ac.uk/gdp/publications/cbn_cbv_lambda.pdf

Ways of eliminating administrative redexes

Administrative redexes can be reduced after the CPS transformation.

• During the translation, mark each λ that corresponds to a source λ.
• After the translation, reduce every redex whose λ is unmarked.

Another idea is to reduce all “no-brainer” redexes. They include the admin.
redexes and are size-decreasing. This can be done on the fly.

Davis, Meehan, Shivers, No-brainer CPS conversion, 2017.

Yet another approach is to define a “one-pass” CPS transformation that
does not produce any administrative redexes in the first place...

https://doi.org/10.1145/3110267

Towards a one-pass transformation

The first step is to make some of the abstractions and applications static.

They should take place at transformation time, not at runtime.

Instead of viewing JtK = λk . . . . as a function of a term to a term,
let us view JtK {w } = . . . as a function of a term and a value to a term.

LxM = x

Lλx .tM = λx .λk . JtK { k }

JvK {w } = w LvM

Jt1 t2K {w } = Jt1K {λx1. Jt2K {λx2. x1 x2 w } }

Jlet x = t1 in t2K {w } = Jt1K {λx . Jt2K {w } }

k denotes a variable; w denotes a value.

Towards a one-pass transformation

This transformation produces fewer administrative redexes:

Jf xK { k } = JfK {λx1. JxK {λx2. x1 x2 k } }
= (λx1. (λx2. x1 x2 k ) x) f

=β (λx2. f x2 k ) x
=β f x k

The remaining administrative redexes arise from the equation

JvK {w } = w LvM

in the case where the continuation w is a λ-abstraction.

How could we alter this equation?

Towards a one-pass transformation

Define the smart application of a (continuation) value w to a value v:

x @β v = x v
(λx .t) @β v = t [v/x]

Note:

• A continuation w is always either a variable or a “transformation” λ,
never a “source” λ, so the redex reduced by w @β v is administrative.

• Provided every “transformation” λ uses its argument linearly, w @β LvM
does not duplicate LvM, so transformed terms remain linear in size.



MPRI 2.4
CPS

François
Pottier

Examples
Interpreter

Traversal

Formulations

Soundness

Remarks

54/94

A one-pass transformation

Change the translation of values. Make every “transformation” λ linear.

LxM = x

Lλx .tM = λx .λk . JtK { k }

JvK {w } = w @β LvM

Jt1 t2K {w } = Jt1K {λx1. Jt2K {λx2. x1 x2 w } }

Jlet x = t1 in t2K {w } = Jt1K {λx . let x = x in Jt2K {w } }

This transformation produces no administrative redexes.

Dargaye and Leroy, Mechanized Verification
of CPS Transformations, 2007.

http://gallium.inria.fr/~xleroy/publi/cps-dargaye-leroy.pdf
http://gallium.inria.fr/~xleroy/publi/cps-dargaye-leroy.pdf

A one-pass transformation

Look Ma, no administrative redexes!

Jf xK { k } = JfK {λx1. JxK {λx2. x1 x2 k } }
= (λx1. (λx2. x1 x2 k ) @β x) @β f
= (λx2. f x2 k ) @β x
= f x k

One drawback of Dargaye and Leroy’s formulation is that · @β · does not
commute with substitutions.

This is repaired in the formulations shown next...

A higher-order formulation

Danvy and Filinski (1992) first defined this one-pass transformation.

Their formulation was in a “higher-order” style.

Let a continuation c be either an arbitrary object or a “transformation” λ:

κ ::= 〈a meta-level function v ⇒ t of values to terms〉
c ::= o w | m κ

Define smart application apply c v and reification reify c as follows:

apply (o w) v = w v – an object-level application
apply (m κ) v = κ(v) – a meta-level application

reify (o w) = w – a no-op
reify (m κ) = λx .(κ(x)) – a “two-level η-expansion”

https://doi.org/10.1017/S0960129500001535

A higher-order formulation

Danvy and Filinski’s transformation is then formulated as follows:

LxM = x
Lλx .tM = λx .λk . JtK {o k }

JvK { c } = apply c LvM
Jt1 t2K { c } = Jt1K {m v1 ⇒ Jt2K {m v2 ⇒ v1 v2 (reify c) } }

Jlet x = t1 in t2K { c } = Jt1K {m v1 ⇒ let x = v1 in Jt2K { c } }

Danvy and Filinski, Representing control:
a study of the CPS transformation, 1992.

Pottier, Revisiting the CPS transformation
and its implementation, 2017.

https://doi.org/10.1017/S0960129500001535
https://doi.org/10.1017/S0960129500001535
http://gallium.inria.fr/~fpottier/publis/fpottier-cps.pdf
http://gallium.inria.fr/~fpottier/publis/fpottier-cps.pdf

A first-order reformulation

Danvy and Filinski’s transformation
can just as well be presented in a “first-order” style.

No need for meta-level functions!

Let us just view m as a binder – roughly, a “transformation” λ:

c ::= o w | mx .t

Define smart application apply c v and reification reify c as follows:

apply (o w) v = w v – an object-level application
apply (mx .t) v = t [v/x] – a meta-level substitution

reify (o w) = w – a no-op
reify (mx .t) = λx .t – a “two-level η-expansion”

A first-order reformulation

Danvy and Filinski’s transformation is then reformulated as follows:

LxM = x
Lλx .tM = λx .λk . JtK {o k }

JvK { c } = apply c LvM
Jt1 t2K { c } = Jt1K {mx1.Jt2K {mx2.x1 x2 (reify c) } }

Jlet x = t1 in t2K { c } = Jt1K {mx1.let x = x1 in Jt2K { c } }

This formulation is simpler than the higher-order formulation.

It is very close to Dargaye and Leroy’s formulation, yet is better behaved: it
commutes with substitution.

A likely reason why Danvy and Filinski did not adopt this formulation is that
their higher-order formulation is closer to an efficient implementation.

The first-order formulation in de Bruijn style

We still view m as a binder:

c ::= o w | m t

Smart application, reification, and substitution c[σ] are as follows:

apply (o w) v = w v – an object-level application
apply (m t) v = t [v/] – a meta-level substitution operation

reify (o w) = w – a no-op
reify (m t) = λt – a two-level η-expansion

(o w)[σ] = o (w[σ]) – apply σ
(m t)[σ] = m (t [⇑ σ]) – apply σ under the binding construct m

The first-order formulation in de Bruijn style

The transformation is formulated in de Bruijn style as follows:

LxM = x
LλtM = λλ(J↑1 tK {o 0 })

JvK { c } = apply c LvM
Jt1 t2K { c } = Jt1K {m J↑1 t2K {m 1 0 ↑2 (reify c) } }

Jlet t1 in t2K { c } = Jt1K {m let 0 in J↑1
1 t2K { ↑2 c } }

↑
i t is short for t [+i]. ↑1

1 t is short for t [⇑ (+1)].

↑
1 can be read as an end-of-scope mark for variable 0.

↑
2 can be read as an end-of-scope mark for variables 0 and 1.

↑
1
1 can be read as an end-of-scope mark for variable 1.

Pottier, Revisiting the CPS transformation
and its implementation, 2017.

http://gallium.inria.fr/~fpottier/publis/fpottier-cps.pdf
http://gallium.inria.fr/~fpottier/publis/fpottier-cps.pdf

1 Examples

From a direct-style interpreter down to an abstract machine

From recursive traversal down to iterative traversal with link inversion

2 Formulations

3 Soundness

4 Remarks

Towards semantic preservation

Let us consider the pure λ-calculus, without “let”.

Let us use de Bruijn notation.

The transformation is defined in CPSDefinition.

The proof of Simulation is in CPSSimulationWithoutLet.

The key lemmas are in CPSSpecialCases, CPSSubstitution, CPSKubstitution.

https://gitlab.inria.fr/fpottier/mpri-2.4-public/tree/master/coq/CPSDefinition.v
https://gitlab.inria.fr/fpottier/mpri-2.4-public/tree/master/coq/CPSSimulationWithoutLet.v
https://gitlab.inria.fr/fpottier/mpri-2.4-public/tree/master/coq/CPSSpecialCases.v
https://gitlab.inria.fr/fpottier/mpri-2.4-public/tree/master/coq/CPSSubstitution.v
https://gitlab.inria.fr/fpottier/mpri-2.4-public/tree/master/coq/CPSKubstitution.v

A small-step simulation diagram

We propose to use the small-step substitution semantics and to establish
a simulation diagram.

One step by the source program is simulated in one or more steps by the
transformed program:

t1 t2

Jt1K { c } Jt2K { c }

cbv

J·K { c } J·K { c }

cbv

+

A solid arrow represents a universal quantification (a hypothesis).
A dashed arrow represents an existential quantification (a conclusion).

Consequences of the simulation diagram

There immediately follows that divergence is preserved.

t1 t2 t3 · · ·

Jt1K { c } Jt2K { c } Jt3K { c } · · ·

cbv

J·K { c }

cbv

J·K { c }

cbv

J·K { c }

cbv

+

cbv

+

cbv

+

The fact that each step is simulated by one or more steps is crucial.

(A proof by co-induction. See Relations/infseq_simulation.)

https://gitlab.inria.fr/fpottier/mpri-2.4-public/tree/master/coq/Relations.v

Consequences of the simulation diagram

Obviously, several steps by the source program
are simulated in several steps by the transformed program:

t1 t2

Jt1K { c } Jt2K { c }

cbv

?

J·K { c } J·K { c }

cbv

?

(A proof by induction. See Relations/star_diamond_left.)

https://gitlab.inria.fr/fpottier/mpri-2.4-public/tree/master/coq/Relations.v

Consequences of the simulation diagram

There follows that convergence to a value is preserved.

We use the identity continuation done, defined as m 0.

t v

JtK {done } JvK {done }

cbv

?

J·K {done } J·K {done }

cbv

?

By definition, JvK {done } is apply done LvM, that is, LvM,
therefore a value.

Thus, the CPS transformation is semantics-preserving.

The simulation lemma

Here is the simulation statement again, this time in textual form:

Lemma (Simulation)

Assume reify c is a value. Then t1 −→cbv t2 implies Jt1K { c } −→+
cbv Jt2K { c }.

Let us now do the proof.

Onscreen or in Coq? Both, probably.

See CPSSimulationWithoutLet.

https://gitlab.inria.fr/fpottier/mpri-2.4-public/tree/master/coq/CPSSimulationWithoutLet.v

Proof of Simulation – case βv

Case: (λt) v −→cbv t [v/]. We must show:

J(λt) vK { c } −→+
cbv Jt [v/]K { c }

By the Value-Value Application lemma, the left-hand term is:

LλtM LvM (reify c)

By definition of LλtM, this is:

(λλ(J↑1 tK {o 0 })) LvM (reify c)

The transformed function is passed an actual argument LvM
and a continuation reify c.



MPRI 2.4
CPS

François
Pottier

Examples
Interpreter

Traversal

Formulations

Soundness

Remarks

70/94

Proof of Simulation – case βv

(λλ(J↑1 tK {o 0 })) LvM (reify c)

In two β-reduction steps, this term reduces to:

(J↑1 tK {o 0 }) [⇑ (LvM/)] [reify c/]

We have two successive substitutions. This term could also be written
using a single substitution that acts on variables 0 and 1:

(J↑1 tK {o 0 }) [reify c · LvM · ids]

(We won’t use this fact, though.)

We now wish to push the substitutions inside, one after the other.

Proof of Simulation – case βv

(J↑1 tK {o 0 }) [⇑ (LvM/)] [reify c/]

By the Substitution lemma, the substitution ⇑ (LvM/)
acts on both the term ↑1 t and the continuation o 0.

However, ⇑ (LvM/) has no effect on variable 0.

Thus, the above term is:

(J(↑1 t)[⇑ (v/)]K {o 0 }) [reify c/]

that is,
(J↑1 t [v/]K {o 0 }) [reify c/]

Proof of Simulation – case βv

(J↑1 t [v/]K {o 0 }) [reify c/]

By the Kubstitution lemma, the substitution reify c/ acts only on the
continuation o 0, not on the term t [v/], because it cancels out with ↑1.

Thus, this term is:
Jt [v/]K { (o 0)[reify c/] }

that is,
Jt [v/]K {o (reify c) }

Proof of Simulation – case βv

We have now reached the term:

Jt [v/]K {o (reify c) }

and the goal is to prove that it reduces (in zero or more steps) to:

Jt [v/]K {o c }

This is the Magic Step lemma. This proof case is finished!

Key Lemmas

Here are the four key lemmas that we have used so far.

Lemma (Value-Value Application)
Jv1 v2K { c } = Lv1M Lv2M (reify c).

Lemma (Substitution)
Let σ and σ′ be value substitutions such that σ′ is equal to σ ; L·M. Then,

(JtK { c })[σ′] = Jt [σ]K { c[σ′] }.

Lemma (Kubstitution)
Let θ and σ be substitutions such that θ ; σ is id. Then,

J(t [θ]K { c })[σ] = JtK { c[σ] }.

Lemma (Magic Step)
JtK {o (reify c) } −→?

cbv JtK { c }.

Proof of Simulation – cases AppL and AppR

Case: t1 u −→cbv t2 u, where t1 −→cbv t2.

We must show Jt1 uK { c } −→+
cbv Jt2 uK { c }.

By definition of the CPS transformation, this is

Jt1K {m J↑1 uK {m 1 0 ↑2 (reify c) } }
−→

+
cbv Jt2K {m J↑1 uK {m 1 0 ↑2 (reify c) } }

Wow – the induction hypothesis applies directly to this goal!

Indeed, reify (m . . .) is a λ-abstraction, therefore a value.

This proof case is complete!

Case: v u1 −→cbv v u2, where u1 −→cbv u2.

Analogous to the previous case, using a Value-Term Application lemma.

We see in these proof cases that reduction under a context in the source
program is translated to reduction at the root in the transformed program.

Simulation in the presence of let constructs

In the presence of “let” constructs, Simulation breaks down.

Challenge: can you find a (minimal) counter-example?

Hint: Enlist a machine’s help. (See next two slides.)



MPRI 2.4
CPS

François
Pottier

Examples
Interpreter

Traversal

Formulations

Soundness

Remarks

77/94

Enumerating λ-terms

Define the size of a term as follows: variables have size 0;
λ-abstractions and applications contribute 1.

Step 1: In OCaml, implement an exhaustive enumeration of the λ-terms of
size s and with at most n free variables. (Given as an exercise in week 1.)

(* Enumerate all variables between 0 and n excluded. *)
let var (n : int) (k : term -> unit) : unit = ...
(* Enumerate all manners of splitting an integer s. *)
let split (s : int) (k : int -> int -> unit) : unit = ...
(* Enumerate all terms of size s with at most n variables. *)
let term (s : int) (n : int) (k : term -> unit) : unit = ...

An enumerator is naturally written in CPS style!

Testing Simulation

Step 2: In OCaml, implement the CPS transformation.

type continuation =
| O of term
| M of term
let cps (t : term) (c : continuation) : term = ...

Step 3: In OCaml, implement a test for the relation · −→?
cbv · :

let reduces (t1 : term) (t2 : term) : bool = ...

Hint: Re-use the auxiliary functions of week 2. See Lambda.

Step 4: Find a term t1 of minimal size that violates Simulation.

Solution: see CPSCounterExample.

https://gitlab.inria.fr/fpottier/mpri-2.4-public/tree/master/ocaml/Lambda.ml
https://gitlab.inria.fr/fpottier/mpri-2.4-public/tree/master/coq/CPSCounterExample.v

Fixing Simulation

In the presence of “let”, Simulation can be fixed as follows:

t1 t2

Jt1K { c } · Jt2K { c }

cbv

J·K { c } J·K { c }

cbv

+

cbv

We allow one step of parallel call-by-value reduction⇒cbv.

The proof of Simulation is more complex; see CPSSimulation.

https://gitlab.inria.fr/fpottier/mpri-2.4-public/tree/master/coq/CPSSimulation.v

Parallel (call-by-value) reduction

Parallel reduction allows reducing all (currently visible) redexes at once,
including under “λ” and in the right-hand side of “let”.

Parallel βv

t1 ⇒cbv t2 v1 ⇒cbv v2

(λt1) v1 ⇒cbv t2[v2/]

Parallel letv
t1 ⇒cbv t2 v1 ⇒cbv v2

let v1 in t1 ⇒cbv t2[v2/]
x ⇒cbv x

t1 ⇒cbv t2
λt1 ⇒cbv λt2

t1 ⇒cbv t2 u1 ⇒cbv u2

t1 u1 ⇒cbv t2 u2

t1 ⇒cbv t2 u1 ⇒cbv u2

let t1 in u1 ⇒cbv let t2 in u2

The ability to reduce under a binder is needed to fix Simulation.

Call-by-name parallel reduction is studied by Takahashi (1995).

Crary (2009) adapts these results to a call-by-value setting.

http://dx.doi.org/10.1006/inco.1995.1057
https://www.cs.cmu.edu/~crary/papers/2009/standard.pdf

Well-behavedness of parallel reduction

· ·

· ·

cbv

+

cbv?

cbv

+

cbv?

Lemma (Commutation)

(⇒?
cbv

; −→+
cbv) ⊆ (−→+

cbv
;⇒?

cbv).

See LambdaCalculusStandardization/pcbv_cbv_commutation.

https://gitlab.inria.fr/fpottier/mpri-2.4-public/tree/master/coq/LambdaCalculusStandardization.v

Well-behavedness of parallel reduction

Lemma (Equiconvergence)

(∃v , t ⇒?
cbv v) ⇐⇒ (∃v ′, t −→?

cbv v ′).

(The idea is, v ′ reduces to v via internal parallel reduction steps.)

See LambdaCalculusStandardization/equiconvergence.

https://gitlab.inria.fr/fpottier/mpri-2.4-public/tree/master/coq/LambdaCalculusStandardization.v

Consequences of Fixed Simulation

There follows that divergence is preserved.

Indeed, from:
t −→cbv · −→cbv · · ·

we get:
JtK { c } −→+

cbv · ⇒cbv · −→
+
cbv · ⇒cbv · · ·

which, by Commutation, yields:

JtK { c } −→+
cbv · −→

+
cbv · ⇒

?
cbv · ⇒cbv · · ·

that is,
JtK { c } −→≥2

cbv · ⇒
?
cbv · · ·

And so on. For an arbitrary n ≥ 0, we have:

JtK { c } −→≥n
cbv · ⇒

?
cbv · · ·

Consequences of Fixed Simulation

Convergence to a value is preserved, too.

Indeed, from:
t −→n

cbv v

we get, as on the previous slide:

JtK {done } −→≥n
cbv · ⇒

?
cbv LvM

and, by Equiconvergence:

∃v ′ JtK {done } −→≥n
cbv · −→

?
cbv v ′

The CPS transformation remains semantics-preserving
in the presence of “let” constructs (phew!).

1 Examples

From a direct-style interpreter down to an abstract machine

From recursive traversal down to iterative traversal with link inversion

2 Formulations

3 Soundness

4 Remarks

Control operators

In a CPS-transformed program, the continuation is a first-class object.

Why not give programmers access to it?

That is, extend the source language with control operators that allow
(delimiting and) capturing the current continuation.

Shift / reset

An example is Danvy and Filinski’s shift / reset (1990).

t ::= . . . | 〈t〉 | ξx .t

A “reset” 〈t〉 does nothing by itself: e.g., 〈42〉 reduces to 42.

A “shift” ξx .t captures the current evaluation context (up to and excluding
the nearest reset), reifies it as a function, and binds the variable x to it.

Then it discards the evaluation context (up to and including the nearest
reset) and executes t instead.

E.g., roughly,

1 + 〈10 + ξc.c (c 100)〉
−→ 1 + (let c = λx .(10 + x) in c (c 100))
−→ 1 + (10 + (10 + 100))
−→ 121

Exercise: Give a small-step semantics to shift / reset.

https://doi.org/10.1145/91556.91622

CPS-transforming shift / reset

The naïve call-by-value CPS transformation is extended as follows:

J〈t〉K = λk .

k (JtK (λy .y))
Jξx .tK = λk . let x = λy .λk ′. k ′ (k y) in

JtK (λy .y)

Exercise (experimental!): Extend the proof of Semantic Preservation.

The target of the transformation is λ-calculus without shift / reset.

It is no longer the case that every call is a tail call, that the right-hand side
of every application is a value, or that continuations are linearly used.

Thus, shift / reset allow reaching terms which previously lied outside the
image of the CPS transformation. CPS lets us think outside the box!

Other control operators

Many other control operators or control constructs can be explained
and compiled away via CPS.

Exceptions can be compiled away by “double-barrelled CPS”,
that is, by using two continuations.

Effect handlers can be compiled away via (type-directed, selective) CPS.

Rompf, Maier, Odersky, Implementing first-class polymorphic delimited
continuations by a type-directed selective CPS-transform, 2009.

Leijen, Type-directed compilation of row-typed algebraic effects, 2017.

See Régis-Gianas’ lectures!

https://doi.org/10.1145/1596550.1596596
https://doi.org/10.1145/1596550.1596596
https://www.microsoft.com/en-us/research/publication/type-directed-compilation-row-typed-algebraic-effects/

Monadic intermediate form

If one just aims to make evaluation order explicit, CPS is overkill.

This transformation, too, achieves indifference:

JxK = x
Jλx .tK = λx .JtK
Jt1 t2K = let x1 = Jt1K in

let x2 = Jt2K in
x1 x2

Jlet x = t1 in t2K = let x = Jt1K in Jt2K

In a transformed term, the components of every application are values.

By further hoisting “let” out of the left-hand side of “let”,
one gets administrative normal form.

Flanagan, Sabry, Felleisen, The essence
of compiling with continuations, 1993 (2003).

https://slang.soe.ucsc.edu/cormac/papers/pldi93.pdf
https://slang.soe.ucsc.edu/cormac/papers/pldi93.pdf
https://www2.ccs.neu.edu/racket/pubs/pldi-fsdf.pdf

The CPS monad

The CPS transformation is a special case of the monadic transformation.

See Dagand’s lectures!

Some history

Continuations, and the CPS transformation, were independently
discovered by many researchers during the 1960s.

John C. Reynolds, The discoveries of continuations, 1993.

http://www.cs.ru.nl/~freek/courses/tt-2011/papers/cps/histcont.pdf

Some history

The CPS transformation has been used in compilers.

Rabbit (Steele). SML/NJ.

Appel, Compiling with Continuations, 1992.

Today, heap-allocating the stack is considered too costly:

• bad locality;
• increased GC load;
• confuses the processor’s built-in prediction of return addresses.

Yet, selective CPS transformations are used to compile effect handlers,

and some compilers use CPS as an intermediate form before coming back
to direct style.

Kennedy, Compiling with continuations, continued, 2007.

http://www.cambridge.org/9780521033114
http://research.microsoft.com/~akenn/sml/CompilingWithContinuationsContinued.pdf

A few things to remember

Continuations rule!

• The CPS transformation achieves several remarkable effects:
• making the stack explicit;
• making evaluation order explicit;
• suggesting/explaining control operators.

• It plays a fundamental role in prog. language theory and in logic.
• Continuation-passing is also a useful programming technique.

We have illustrated a few proof techniques:

• A small-step simulation diagram, in a proof of semantic preservation.
• Testing, to refute a conjecture and find a counter-example!

